    ПРИЛОЖЕНИЕ 4.  БИБЛИОТЕКА  МОДЕЛЕЙ ПОЛУПРОВОДНИКОВЫХ ПРИБОРОВ





        В настоящее время составлена библиотека математических моделей оте-


чественных диодов,  транзисторов, операционных усилителей и компарато-


ров напряжения,  включающая в себя около 300 наименований.  В связи  с


тем,  что публикация ее содержания в книге не имеет смысла из-за боль-


шого труда по  переносу на магнитные носители,  приведем здесь  лишь


ее каталог и примеры ряда типичных моделей. В полном объеме библиотека


может быть поставлена на дискетах.





                     Диоды выпpямительные





 2Д103А                        		KД102A      		Д237А, Б, В, Г, Е


 2Д104А                       		KД103A      		КД2997


 2Д106А                       	 	KД104A      		КД503А


 2Д203А, Б, В, Г, Д            		KД105A      		КД509А


 2Д212А, Б, В, Г               		КД202Д      		KД510A


 2Д213А, Б, В, Г               		КД203А      		КД512А


 2Д220А, Б, В, Г, Д, Е, Ж, И  	КД204А     		КД513А


 2Д419А, Б, В                 		КД208А      		КД520А


 2Д420А                        		КД209А      		КД521А


 2Д503Б                        		KД212A      		KД522А, Б


 2Д510А                       		KД213A, Б


 2Д521А                        		КД219А


 2Д522А, Б                     		КД220А





.model D2D104A D (Is=162n N=2.887 Rs=1.031 Ikf=1.82 Xti=3 Eg=1.11


+    Cjo=29.58p M=.3333 Vj=.75 Fc=.5 Isr=34.09E-30 Nr=5.963 Bv=500.2


+    Ibv=.2589 Tt=5.771u)


*


.model KD209A  D  (IS=6.22e-11 N=1.23 RS=0.17 CJO=16.2 TT=7.21e-7


+    M=0.35 VJ=0.68 FC=0.5 BV=600 IBV=1e-10 EG=1.11 XTI=3)


*


.model KD509A  D  (IS=1.14e-12 N=1.12 RS=2.1 CJO=2.15p TT=3.25n


+    M=0.33 VJ=0.71 FC=0.5 BV=50 IBV=1E-11 EG=1.11 XTI=3)


*


.model KD522B  D  (Is=33.85e-21 Rs=0 N=1 Xti=3 Eg=1.11 Bv=30 Ibv=2u


+    Cjo=4p Vj=.75 M=.3333 Fc=.5 Tt=4p)





                          Стабилитроны





Д814А, Б, В, Г, Д             		2С211Г


Д815А, Б, В, Г, Д, Е, Ж        	2С522А


Д816А, Б, Г, Д                 		2С527А


Д817А, Б, В, Г                  		КС133А


Д818Е                           		КС139А


2С133А                           		КС147А


2С147А                           		КС156А


2С168А                           		КС168А


2С191Г





.model D814A D(IS=.3920e-12 N=1.19 RS=1.25 CJO=41.15p  TT=49.11n


+    M=0.41 VJ=0.73 FC=0.5 BV=8 IBV=0.5u EG=1.11 XTI=3)


*


.model D818e D(Is=3m Rs=18 N=1 Xti=3  Eg=1.11  Bv=9  Ibv=3m Cjo=1p


+    Vj=.75 M=.3333 Fc=.5 Tt=5n)


*


.model KS133A D(IS=89.00e-15 N=1.16 RS=25  CJO=72.00p  TT=57.00n


+    M=0.47 VJ=0.80 FC=0.5 BV=3.3 IBV=5u EG=1.11 XTI=3)


*


.model 2S527A D(IS=25.00e-15 N=1.11  RS=4  CJO=47.42p  TT=256.6n


+    M=0.39 VJ=0.68 FC=0.5 BV=27 IBV=5u EG=1.11 XTI=3)





                    Биполярные транзисторы





p(n(p            		 n(p(n            		p(n(p           		n(p(n





2Т202А, Б, В, Г  	2Т215А, Б, В, Г   	Т208К


2Т203Д           	2T312A, Б, B      	КТ351А


2Т214А, Б, В, Г  	2Т315А, Е, Д, И   	КТ363А


2Т313Б           	2Т316Б, Д         	КТ364А, Б, В


2Т326Б           	2Т324А, Б, В,     	КТ375А, Б


2Т360А, Б, Г          		 Г, Д, Е   	ТС3103А-1,       	КТ312Б


2Т363Б           	2Т325Б            	КТ3107А,Б,В,Г,


2Т364А-2, Б-2    	2Т354А              	Д,Е,ГЕ,И,К,Л   	КТ315А,Б,В,Г,


	В-2     		2Т355А                               				Д,Е,ГЕ,И


2Т370А-1         	2Т368Б            	КТ830А, Б, В, Г  	КТ316b


2Т3107А, Е, И, 	К 2Т371А                             		КТ385АМ-2


2Т3108А, Б, В    	2Т384AM-2         	КТ836А, Б, В     	КТ399А


2Т626А           	2Т385А                             			КТ3102А, Б, БМ,


2Т632А           	2Т388А                                       			В, Г, Е


2Т639Б           	2Т397                              			КТ3117А


2Т644Б           	2Т399А                             			КТ608А, Б


2Т708            		2Т3102А, Б, Е                      		КТ630А, Б, В,


2Т709            		2Т3117А                                   			Г, Д, Е


2Т818А           	2Т504                              			КТ635Б


2Т825А           	2Т505                              			КТ642А-2


2Т830А           	2Т602А, Б                          			КТ653Б


                 		2Т625А, Б                          			КТ815Г


                 		2Т629А                             			КТ819А, Б, В


                 		2Т630А, Б, В,                      		КТ831А, Б, В


                         		Г, Д                       			КТ841А


                 		2Т638А                             			КТ847А


                 		2Т646А                             			КТ866А


                 		2Т819А, Б, В, Г                    		КТ880А


                 		2Т827А                             			КТ926А


                 		2Т831А


                 		2Т921А


                 		2T939A





.model 2t312a  NPN  (Is=21f  Xti=3  Eg=1.11   Vaf=126.2   Bf=86.76


+   Ise=189f Ne=1.328   Ikf=.164   Nk=.5  Xtb=1.5  Br=1  Isc=66.74f


+   Nc=1.385 Ikr=1.812  Rb=300  Rc=.897  Cjc=8.4p Mjc=.29  Vjc=.692


+   Fc=.5 Cje=26.53p Mje=.333 Vje=.75 Tr=10n Tf=1.743n Itf=1)


*


.model 2T315A NPN (Is=23.68f Xti=3 Eg=1.11 Vaf=60 Bf=108  Ne=1.206


+   Ise=23.68f Ikf=.1224 Xtb=1.5 Br=4.387G Nc=1.8  Isc=900p Ikr=20m


+   Rc=5  Cjc=7p  Vjc=.7  Mjc=.333  Fc=.5  Cje=10p  Vje=.7 Mje=.333


+   Tr=130.5n Tf=1n Itf=40m Vtf=80 Xtf=1.1)


*


.model 2t371a  NPN  (Is=1.378f  Xti=3  Eg=1.11  Vaf=68.25   Bf=236


+   Ne=1.479 Ise=43.8f  Ikf=.1777   Xtb=1.5  Var=45  Br=3.414  Nc=2


+   Isc=55f Ikr=35m Rb=44.1 Rc=2.8 Cjc=1.932p Vjc=.75 Mjc=.33 Fc=.5


+   Cje=1.747p Vje=.69 Mje=.33  Tr=13.65n  Tf=43.78p Itf=.35 Vtf=10


+   Xtf=2)


*


.model 2T3102A  NPN (Is=685.1f Xti=3 Eg=1.11 Vaf=80 Bf=213 Ne=2.55


+   Ise=563.4p Ikf=.43 Xtb=1.5 Br=11.68G Nc=2 Isc=100p Ikr=80m Rc=7


+   Cjc=12.07p Vjc=.7 Mjc=.333 Fc=.5  Cje=9.284p  Vje=.75  Mje=.333


+   Tr=80p Tf=252.6p Itf=8m Vtf=50 Xtf=1.2)


*


.model KT3117A  NPN  (IS=10e-15  ISE=1uA  NE=4  ISC=1uA NC=4 BF=200


+   IKF=.4A VAF=70 CJC=10pF CJE=30pF RB=5 RE=.5 TF=0.6NS TR=200ns)


*


.model KT608A  NPN  (IS=1e-12 VAF=60 VAR=8V BF=80 IKF=.4A NC=4 NE=4


+   RB=3 RE=.5 ISE=10uA ISC=10uA CJC=12pF CJE=30pF TR=50NS TF=.6ns)


*


.model KT608B  NPN (IS=1e-12 VAF=60 VAR=8V BF=160 IKF=.4A NC=4 NE=4


+   RB=3 RE=.5 ISE=10uA ISC=10uA CJC=12pF CJE=30pF TR=50ns TF=.6ns)


*


.model 2T630A NPN (Is=442.7e-18 Xti=3 Eg=1.11 Vaf=120 Bf=80 Ne=1.5


+  Ise=1e-7 Ikf=.8  Xtb=1.5  Br=1 Nc=2 Isc=1E-6 Ikr=.7 Rc=0 Cjc=25p


+  Vjc=.75 Mjc=.333 Fc=.5 Cje=65p Vje=.75 Mje=.333 Tr=10n Tf=3.129n)


*


.model KT830A PNP  (Is=141f  Xti=3  Eg=1.11  Vaf=100  Bf=25  Ne=1.5


+   Ise=100u Ikf=1.5 Xtb=1.5 Br=1 Nc=2 Isc=100u Ikr=0 Rc=0 Cjc=180p


+   Vjc=.75 Mjc=.333 Fc=.5   Cje=210p   Vje=.75   Mje=.333   Tr=10n


+   Tf=31.77n Itf=1.5 Vtf=100 Xtf=1.3)


*


.model 2T939a  NPN  (Is=4.08f  Xti=3  Eg=1.11  Vaf=25.48  Bf=105.4


+  Ise=56.88f Ne=1.39  Ikf=1.763  Nk=.82  Xtb=1.5  Br=1  Isc=365.6f


+  Nc=1.825  Ikr=.398  Rb=6  Rc=.818  Cjc=10.12p Mjc=.8707 Vjc=7.26


+  Fc=.5 Cje=17.53p Mje=.4951 Vje=1.472 Tr=10n Tf=40.66p  Itf=10.44


+  Xtf=70.42 Vtf=10)





                Макромодели транзисторных сборок





                     p-n-p             		n-p-n


                  198НТ5, 6, 7, 8   	129НТ1, Б


                  КТС393Б           	159НТ1


                  КТС622А           	198НТ1, 2, 3, 4


                  2ТС622А, Б        	1НТ251А


                                    		КТС398А, Б





*---------------------------  198NT1 --------------------------


*                                               					  E1


*              		E3  B3  C3 E5  B5  C5  E4  B4  C4  C2  B2  E2  B1  C1


*              		 |      |      |     |      |     |      |      |      |      |     |      |      |     |


.SUBCKT  198NT1   1    2    3    4     5    6     7     8     9    10  11   12   13  14


Q1   14   13   12   NT1


Q2   10   11   12   NT1


Q3     3     2     1   NT1


Q4     9     8     7   NT1


Q5     6     5     4   NT1


.model NT1 NPN (Is=33.67f Xti=3 Eg=1.11 Vaf=100 Bf=150 Ne=1.5 Ise=0


+   Ikf=20E-3 Xtb=1.5 Br=1  Nc=2  Isc=0  Ikr=0  Rc=0 Cjc=7p Vjc=.75


+   Mjc=.333 Fc=.5 Cje=25p Vje=.75 Mje=.3333 Tr=10n Tf=158.9n Itf=0


+   Vtf=0 Xtf=0)


.ENDS


*----------------------------  KTC398A  -----------------------


*                                      E1   C1   B1  B2  C2   E2


*                 		        |      |       |      |      |      |


.SUBCKT   KTC398A    1     2     3     4     5     6


Q1     2     3     1     NT1


Q2     5     4     6     NT1


.model NT1  NPN  (Is=5f Xti=3 Eg=1.12 Vaf=20 Bf=150 Ne=1.2 Ise=750f


+   Ikf=25m Nc=2 Xtb=1.5  Br=2  Isc=650f  Ikr=20m  Rc=50  Cjc=2.96p


+   Vjc=.75 Mjc=.333 Fc=.5  Cje=2.653p  Vje=.75  Mje=.333  Tr=1.12n


+   Tf=68.3p Itf=.02 Vtf=20 Xtf=1.1)


.ENDS





                        Полевые транзисторы





.model  J2PS104G NJF(Vto=-1.562 Beta=816.3u Lambda=8.83m Rs=31 Rd=31


+  Cgs=3.2p Cgd=2.2p Fc=0.5 Pb=1 Is=10f)


*


.model  J2P303B NJF(Vto=-0.8137 Beta=2.015m Lambda=17.89m Rs=44 Rd=44


+  Cgs=4.2p Cgd=3.8p Fc=0.5 Pb=1 Is=10f)


*


.model  J2P307B NJF(Vto=-2.617 Beta=1.578m Lambda=1.890m Rs=15 Rd=15


+  Cgs=3.5p Cgd=3p Fc=0.5 Pb=1 Is=10f)











  ПРИЛОЖЕНИЕ 5. БИБЛИОТЕКА МОДЕЛЕЙ ЦИФРОВЫХ И АНАЛОГО-ЦИФРОВЫХ ИС





   Рассмотрим характерные  особенности составления моделей цифровых и аналого-цифровых ИС для программы PSpice на нескольких конкретных примерах [24, 28].


�Рис. П5.1. Логический элемент 2И(НЕ


   1. Логический ТТЛ(элемент 2И(НЕ 133ЛА3.  Модели  всех  ИС,  даже простейших, необходимо оформлять в виде макромоделей (subckt). В связи с тем, что для однородных ИС составляются символы  одной  секции,  ниже приведем  описание секции ИС 133ЛА (рис.  П5.1):





.subckt    A   B   NQ


U1  NAND(2)  $G_DPWR  $G_DGND   A   B   NQ   D_133   IO_133


.ends





   Здесь 133LA3 ( имя макромодели,  совпадающее с именем SYM-файла; $G_DPWR,  $G_DGND ( имена узлов питания и цифровой “земли”; A,  B ( имена входов секции; NQ ( имя выхода; U1 ( имя компонента в составе макромодели;  NAND ( имя стандартной модели вентиля И(НЕ; в круглых скобках указано общее количество входов в корпусе ( 2;  далее перечисляются имена узлов подключения макромодели;  D_133 ( имя модели динамики вентиля 133ЛА3; IO_133 ( модель входа/выхода вентиля.


   Модель входа/выхода имеет вид:





.model   IO_133   UIO (drvh=50 drvl=50 AtoD=AtoD_133, DtoA=DtoA_133)





   Здесь UIO  ( cтандартное ключевое слово;  drvh,  drvl ( выходные сопротивления вентиля в состоянии высокого и низкого  уровня  соответственно (используются при определении логического уровня сигнала на входе ИС,  к которому параллельно подключены  выходы  нескольких источников  сигнала);  AtoD,  DtoA ( имена моделей интерфейсов аналог/цифра (А/Ц) и цифра/аналог (Ц/А) соответственно.


   Модель динамики вентиля имеет вид





.model   D_133   UGATE (tplhty=2.2n, tplhmx=5n, tphlty=1.5n, tphlmx=3n)





   Здесь UGATE ( стандартное ключевое слово; в скобках указаны значения задержек при переходе от низкого уровня к высокому и от высокого уровня к низкому (для каждой задержки может быть указано минимальное, типичное и максимальное значения).


   Модель интерфейса А/Ц имеет вид





.subckt AtoD_133   A   D


o1   A   $G_DGN  D IN_133   Dgtlnet=D IO_133


.ends





   Модели интерфейса А/Ц и Ц/А включаются в схему между аналоговыми и цифровыми узлами автоматически. Здесь A ( имя аналогового узла; D ( имя цифрового узла;  $G_DGND ( глобальное имя “земли”;  IN_133  ( имя модели интерфейса А/Ц.


   Модель интерфейса А/Ц имеет вид





.model   IN_133   DOUTPUT(


+    s0name=0, s0vlo=-10, s0vhi=.5,


+    s1name=1, s1vlo=2.5, s1vhi=10,


+    s2name=x, s2vlo=.5,  s2vhi=2.5)





   Здесь DOUTPUT ( ключевое слово;  s0name,  s1name, s2name ( имена логических состояний “0”,  “1” и неопределенного состояния “X”; для каждого состояния указываются граничные значения напряжений.


   Модель интерфейса Ц/А имеет вид





.subckt DtoA_133 D A


n1 A $G_DGND $G_DPWR OUT_133 Dgtlnet=D IO_133


.ends





   Здесь $G_DPWR ( глобальное имя цепи питания; OUT_133 ( имя модели интерфейса Ц/А.


   Модель интерфейса Ц/А имеет вид





.model OUT_133 DINPUT (


+ s0name=0, s0tsw=3.5n,  s0rlo=7,    s0rhi=390,


+ s1name=1, s1tsw=5.5n,  s1rlo=467,  s1rhi=200,


+ s2name=x, s2tsw=3.5n,  s2rlo=43,   s2rhi=106,


+ s3name=z, s3tsw=3.5n,  s3rlo=200k, s3rhi=200k)





   Здесь DINPUT ( ключевое слово;  s0name, s1name, s2name, s3name ( имена логических состояний “0”,  “1”, неопределенного состояния “X” и состояния высокого импеданса “Z”;  после имени каждого  состояния указывается значение времени переключения и сопротивления между выходным узлом и “землей” и источником питания.


   Источник питания +5 В,  подключаемый к интерфейсу Ц/А автоматически, имеет модель вида





.subckt   DIGIFPWR    GND


VDPWR   	$G_DPWR 	$G_DGND 	5v


R1  		$G_DPWR  	GND 		1MEG


VDGND 	$G_DGND 	GND     	0v


R2    		$G_DGND 	GND     	1MEG


.ends





   Модель источника  питания включается в библиотеку моделей цифровых компонентов один раз.


   2.  Логический  ТТЛ(элемент 2И(НЕ с открытым коллекторным выходом 133ЛА8 (рис. П5.2). По сравнению с моделью вентиля 133ЛА3 этот элемент имеет другие значения сопротивлений в состоянии “1”





.subckt   133LA8   A   B   NQ


U1   NAND(2)   $G_DPWR   $G_DGND   A   B   NQ    D_133   IO_133_C


.ends


�Рис. П5.2. Логический элемент 2И(НЕ с открытым коллектором





 Модель входа/выхода имеет вид


.model IO_133_C UIO (drvh=50 drvl=50 AtoD=AtoD_133_C DtoA=DtoA_133_C)





Модель интерфейса А/Ц имеет вид





.subckt   AtoD_133_C   A   D


o1   A   $G_DGND   IN_133   Dgtlnet=D IO_133_C


.ends





Модель интерфейса Ц/А имеет вид





.subckt   DtoA_133_C    D    A


n1    A    $G_DGND    $G_DPWR    OUT_133_C   Dgtlnet=D IO_133_C


.ends





Модель интерфейса Ц/А имеет вид





.model    OUT_133_C    DINPUT (


+ s0name=0, s0tsw=3.5n,  s0rlo=7,    s0rhi=390,


+ s1name=1, s1tsw=5.5n,  s1rlo=200k, s1rhi=200k,


+ s2name=x, s2tsw=3.5n,  s2rlo=43,   s2rhi=106,


+ s3name=z, s3tsw=3.5n,  s3rlo=200k, s3rhi=200k)





   3. Двойной  4-входовый  коммутатор 564КП1.  Условное графическое отображение и функциональная схема  ИС  564КП1  приведены  на  рис. П5.3. Ниже представим его макромодель на языке программы PSpice:





.subckt  564KP1   V   A0   A1   X1   X2   X3   X4   Y1   Y2   Y3   Y4    X    Y


U1  BUFA(2)  $G_DPWR  $G_DGND  A0  A1   B0    B1   KP1SLCT_HC  IO_HC


U2  INVA(2)   $G_DPWR  $G_DGND  A0  A1   IB0  IB1  KP1SLCT_HC  IO_HC


U3  NORA(3,4)  $G_DPWR  $G_DGND   B0   B1   V     IB0   B1   V


+             B0    IB1   V      IB0  IB1   V


+            1Q0  1Q1  1Q2  1Q3  KP1MLTI_HC   IO_HC


U4   NORA(3,4)   $G_DPWR   $G_DGND    B0   B1   V     IB0   B1   V


+            B0  IB1   V     IB0  IB1   V


+           2Q0  2Q1  2Q2  2Q3  KP1MLTI_HC IO_HC


S1  X1  X  1Q0  0     K1


S2  X2  X  1Q1  0     K1


S3  X3  X  1Q2  0     K1


S4  X4  X  1Q3  0     K1


S5  Y1  Y  2Q0  0     K1


S6  Y2  Y  2Q1  0     K1


S7  Y3  Y  2Q2  0     K1


S8  Y4  Y  2Q3  0     K1


C1  X1  0  10p


C2  X2  0  10p


C3  X3  0  10p


C4  X4  0  10p


C5  Y1  0  10p


C6  Y2  0  10p


C7  Y1  0  10p


C8  Y2  0  10p


C9   X  0  10p


C10  Y  0  10p


.ends 564KP1


.model KP1SLCT_HC  UGATE


+       (tplhty=4ns     tplhmx=3ns      tphlty=4ns      tphlmx=3ns)


.model KP1STRB_HC  UGATE


+       (tplhty=11ns    tplhmx=24ns     tphlty=11ns     tphlmx=24ns)


.model KP1MLTI_HC  UGATE


+       (tplhty=17ns    tplhmx=35ns     tphlty=17ns     tphlmx=35ns)


.model K1 VSWITCH (VON=7.5 VOFF=2.8)


�Рис. П5.3. Коммутатор 564КП1





   4. 12-разрядный ЦАП К594ПА1. Условное графическое обозначение и упрощенная функциональная схема ИС К594ПА1 приведены на рис.  П5.4. Разряды преобразуемых чисел подаются на входы 18,  17, ..., 7 и через буферы U1, ..., U12 коммутируют источники взвешенных токов G1, ..., G12. Источник постоянного тока IERR имитирует погрешность преобразования 0,2 мкА. Суммирование тока осуществляется с помощью управляемого источника тока GOUT, максимальный выходной ток равен 2,2 мА. Задержка преобразования имитируется с помощью задержек  буферов U1, ..., U12. Ниже приведено текстовое описание модели.





.subckt  594PA1   1   2   3   4   7   8   9   10   11   12   13   14   15   16   17   18   23


R21  1  2  9.95k


R19  3  4  5k


REF  23  0  5k


G1   30 0 VALUE= {V(1R)}


G2   30 0 VALUE= {V(2R)*0.5}


G3   30 0 VALUE= {V(3R)*0.25}


G4   30 0 VALUE= {V(4R)*0.125}


G5   30 0 VALUE= {V(5R)*0.0625}


G6   30 0 VALUE= {V(6R)*0.03125}


G7   30 0 VALUE= {V(7R)*0.015625}


G8   30 0 VALUE= {V(8R)*0.0078125}


G9   30 0 VALUE= {V(9R)*0.00390625}


G10  30 0 VALUE={V(10R)*0.001953125}


G11  30 0 VALUE={V(11R)*0.0009765625


G12  30 0 VALUE={V(12R)*0.00048828125


IERR 30 0 0.2uA


RI   30 0 1.0


GOUT  3 0 VALUE={V(30)*1.1mA}


R1   1R   0   1MEG


R2   2R   0   1MEG


R3   3R   0   1MEG


R4   4R   0   1MEG


R5   5R   0   1MEG


R6   6R   0   1MEG


R7   7R   0   1MEG


R8   8R   0   1MEG


R9   9R   0   1MEG


R10 10R  0  1MEG


R11 11R  0  1MEG


R12 12R  0  1MEG


U1   BUF   $G_DPWR   $G_DGND  18   1R   DN IO_N


U2   BUF   $G_DPWR   $G_DGND  17   2R   DN IO_N


U3    BUF   $G_DPWR   $G_DGND  16   3R   DN IO_N


U4    BUF   $G_DPWR   $G_DGND  15   4R   DN IO_N


U5    BUF   $G_DPWR   $G_DGND  14   5R   DN IO_N


U6    BUF   $G_DPWR   $G_DGND  13   6R   DN IO_N


U7    BUF   $G_DPWR   $G_DGND  12   7R   DN IO_N


U8    BUF   $G_DPWR   $G_DGND  11   8R   DN IO_N


U9    BUF   $G_DPWR   $G_DGND  10   9R   DN IO_N


U10  BUF   $G_DPWR   $G_DGND   9  10R   DN IO_N


U11  BUF   $G_DPWR   $G_DGND   8  11R   DN IO_N


U12  BUF   $G_DPWR   $G_DGND   7  12R   DN IO_N


.ends 594PA1


*


.model   DN   UGATE


.model   IO_N   UIO (drvh=0 drvl=0 AtoD=AtoD_N DtoA=DtoA_N)


.subckt   AtoD_N   A   D


o1   A   $G_DGND   DOUT_N Dgtlnet=D IO_N


.ends


.subckt   DtoA_N   D   A


n1   A   $G_DGND   $G_DPWR   DIN_N   dgtlnet=d IO_N


.ends


.model   DOUT_N   DOUTPUT (


+  s0name=0, s0vlo=-100, s0vhi=0.5,


+  s1name=1, s1vlo=2.5,  s1vhi=100,


+  s2name=x, s2vlo=0.5,  s2vhi=2.5)


.model   DIN_N   DINPUT (


+  s0name=0, s0tsw=1n, s0rlo=1,   s0rhi=1k,


+  s1name=1, s1tsw=1n, s1rlo=200, s1rhi=800,


+  s2name=x, s2tsw=1n, s2rlo=100, s2rhi=900,


+  s3name=z, s3tsw=1n, s3rlo=200, s3rhi=800)


�Рис. П5.4. 12-разрядный ЦАП  К594ПА1





В приведенных фрагментах библиотеки моделей цифровых ИС не было необходимости вводить устройства контроля времени установки/удержания. С ними можно познакомиться, например, в библиотеке ТТЛ(логики 7400.slb в моделях декодера 7442A, счетчика 7490A или сдвигающего регистра 7491А.

